# NAG Library Routine Document

# F08JBF (DSTEVX)

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

## 1 Purpose

F08JBF (DSTEVX) computes selected eigenvalues and, optionally, eigenvectors of a real symmetric tridiagonal matrix $A$. Eigenvalues and eigenvectors can be selected by specifying either a range of values or a range of indices for the desired eigenvalues.

## 2 Specification

```
SUBROUTINE F08JBF (JOBZ, RANGE, N, D, E, VL, VU, IL, IU, ABSTOL, M, W, Z,       &
                   LDZ, WORK, IWORK, JFAIL, INFO)

INTEGER            N, IL, IU, M, LDZ, IWORK(5*N), JFAIL(*), INFO
REAL (KIND=nag_wp) D(*), E(*), VL, VU, ABSTOL, W(N), Z(LDZ,*), WORK(5*N)
CHARACTER(1)       JOBZ, RANGE
```

The routine may be called by its LAPACK name ***dstevx***.

## 3 Description

F08JBF (DSTEVX) computes the required eigenvalues and eigenvectors of $A$ by reducing the tridiagonal matrix to diagonal form using the $QR$ algorithm. Bisection is used to determine selected eigenvalues.

## 4 References

Anderson E, Bai Z, Bischof C, Blackford S, Demmel J, Dongarra J J, Du Croz J J, Greenbaum A, Hammarling S, McKenney A and Sorensen D (1999) *LAPACK Users' Guide* (3rd Edition) SIAM, Philadelphia http://www.netlib.org/lapack/lug

Demmel J W and Kahan W (1990) Accurate singular values of bidiagonal matrices *SIAM J. Sci. Statist. Comput.* **11** 873–912

Golub G H and Van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

## 5 Parameters

1: JOBZ – CHARACTER(1) *Input*

*On entry*: indicates whether eigenvectors are computed.

JOBZ = 'N'
Only eigenvalues are computed.

JOBZ = 'V'
Eigenvalues and eigenvectors are computed.

*Constraint*: JOBZ = 'N' or 'V'.

2: RANGE – CHARACTER(1) *Input*

*On entry*: if RANGE = 'A', all eigenvalues will be found.

If RANGE = 'V', all eigenvalues in the half-open interval $(\mathrm{VL}, \mathrm{VU}]$ will be found.

If RANGE = 'I', the ILth to IUth eigenvalues will be found.

*Constraint*: RANGE = 'A', 'V' or 'I'.

3: N – INTEGER *Input*

*On entry*: $n$, the order of the matrix.

*Constraint*: $N \geq 0$.

4: D($*$) – REAL (KIND=nag_wp) array *Input/Output*

**Note**: the dimension of the array D must be at least $\max(1, N)$.

*On entry*: the $n$ diagonal elements of the tridiagonal matrix $A$.

*On exit*: may be multiplied by a constant factor chosen to avoid over/underflow in computing the eigenvalues.

5: E($*$) – REAL (KIND=nag_wp) array *Input/Output*

**Note**: the dimension of the array E must be at least $\max(1, N - 1)$.

*On entry*: the $(n - 1)$ subdiagonal elements of the tridiagonal matrix $A$.

*On exit*: may be multiplied by a constant factor chosen to avoid over/underflow in computing the eigenvalues.

6: VL – REAL (KIND=nag_wp) *Input*
7: VU – REAL (KIND=nag_wp) *Input*

*On entry*: if RANGE = 'V', the lower and upper bounds of the interval to be searched for eigenvalues.

If RANGE = 'A' or 'I', VL and VU are not referenced.

*Constraint*: if RANGE = 'V', VL < VU.

8: IL – INTEGER *Input*
9: IU – INTEGER *Input*

*On entry*: if RANGE = 'I', the indices (in ascending order) of the smallest and largest eigenvalues to be returned.

If RANGE = 'A' or 'V', IL and IU are not referenced.

*Constraints*:

if RANGE = 'I' and N = 0, IL = 1 and IU = 0;
if RANGE = 'I' and N > 0, $1 \leq IL \leq IU \leq N$.

10: ABSTOL – REAL (KIND=nag_wp) *Input*

*On entry*: the absolute error tolerance for the eigenvalues. An approximate eigenvalue is accepted as converged when it is determined to lie in an interval $[a, b]$ of width less than or equal to

$$\mathrm{ABSTOL} + \epsilon \max(|a|, |b|),$$

where $\epsilon$ is the ***machine precision***. If ABSTOL is less than or equal to zero, then $\epsilon \|A\|_1$ will be used in its place. Eigenvalues will be computed most accurately when ABSTOL is set to twice the underflow threshold $2 \times$ X02AMF( ), not zero. If this routine returns with INFO > 0, indicating that some eigenvectors did not converge, try setting ABSTOL to $2 \times$ X02AMF( ). See Demmel and Kahan (1990).

11: M – INTEGER *Output*

*On exit*: the total number of eigenvalues found. $0 \leq M \leq N$.

If RANGE = 'A', M = N.

If RANGE = 'I', M = IU − IL + 1.

12: W(N) – REAL (KIND=nag_wp) array *Output*

*On exit*: the first M elements contain the selected eigenvalues in ascending order.

13: Z(LDZ,∗) – REAL (KIND=nag_wp) array *Output*

**Note**: the second dimension of the array Z must be at least $\max(1, \mathrm{M})$ if JOBZ = 'V', and at least 1 otherwise.

*On exit*: if JOBZ = 'V', then

if INFO = 0, the first M columns of $Z$ contain the orthonormal eigenvectors of the matrix $A$ corresponding to the selected eigenvalues, with the $i$th column of $Z$ holding the eigenvector associated with $\mathrm{W}(i)$;

if an eigenvector fails to converge (INFO > 0), then that column of $Z$ contains the latest approximation to the eigenvector, and the index of the eigenvector is returned in JFAIL.

If JOBZ = 'N', Z is not referenced.

**Note:** you must ensure that at least $\max(1, \mathrm{M})$ columns are supplied in the array Z; if RANGE = 'V', the exact value of M is not known in advance and an upper bound of at least N must be used.

14: LDZ – INTEGER *Input*

*On entry*: the first dimension of the array Z as declared in the (sub)program from which F08JBF (DSTEVX) is called.

*Constraints*:

if JOBZ = 'V', $\mathrm{LDZ} \geq \max(1, \mathrm{N})$;
otherwise $\mathrm{LDZ} \geq 1$.

15: WORK($5 \times \mathrm{N}$) – REAL (KIND=nag_wp) array *Workspace*

16: IWORK($5 \times \mathrm{N}$) – INTEGER array *Workspace*

17: JFAIL(∗) – INTEGER array *Output*

**Note**: the dimension of the array JFAIL must be at least $\max(1, \mathrm{N})$.

*On exit*: if JOBZ = 'V', then

if INFO = 0, the first M elements of JFAIL are zero;

if INFO > 0, JFAIL contains the indices of the eigenvectors that failed to converge.

If JOBZ = 'N', JFAIL is not referenced.

18: INFO – INTEGER *Output*

*On exit*: INFO = 0 unless the routine detects an error (see Section 6).

## 6 Error Indicators and Warnings

Errors or warnings detected by the routine:

INFO < 0

If INFO = $-i$, argument $i$ had an illegal value. An explanatory message is output, and execution of the program is terminated.

INFO $> 0$

If INFO $= i$, then $i$ eigenvectors failed to converge. Their indices are stored in array JFAIL. Please see ABSTOL.

## 7 Accuracy

The computed eigenvalues and eigenvectors are exact for a nearby matrix $(A + E)$, where

$$\|E\|_2 = O(\epsilon)\|A\|_2,$$

and $\epsilon$ is the ***machine precision***. See Section 4.7 of Anderson *et al.* (1999) for further details.

## 8 Further Comments

The total number of floating point operations is proportional to $n^2$ if JOBZ = 'N' and is proportional to $n^3$ if JOBZ = 'V' and RANGE = 'A', otherwise the number of floating point operations will depend upon the number of computed eigenvectors.

## 9 Example

This example finds the eigenvalues in the half-open interval $(0, 5]$, and the corresponding eigenvectors, of the symmetric tridiagonal matrix

$$A = \begin{pmatrix} 1 & 1 & 0 & 0 \\ 1 & 4 & 2 & 0 \\ 0 & 2 & 9 & 3 \\ 0 & 0 & 3 & 16 \end{pmatrix}.$$

### 9.1 Program Text

```
    Program f08jbfe

!     F08JBF Example Program Text

!     Mark 24 Release. NAG Copyright 2012.

!     .. Use Statements ..
      Use nag_library, Only: dstevx, nag_wp, x02amf, x04caf
!     .. Implicit None Statement ..
      Implicit None
!     .. Parameters ..
      Integer, Parameter               :: nin = 5, nout = 6
!     .. Local Scalars ..
      Real (Kind=nag_wp)               :: abstol, vl, vu
      Integer                          :: ifail, il, info, iu, ldz, m, n
!     .. Local Arrays ..
      Real (Kind=nag_wp), Allocatable  :: d(:), e(:), w(:), work(:), z(:,:)
      Integer, Allocatable             :: iwork(:), jfail(:)
!     .. Executable Statements ..
      Write (nout,*) 'F08JBF Example Program Results'
      Write (nout,*)
!     Skip heading in data file
      Read (nin,*)
      Read (nin,*) n
      ldz = n
      m = n
      Allocate (d(n),e(n),w(n),work(5*n),z(ldz,m),iwork(5*n),jfail(n))

!     Read the lower and upper bounds of the interval to be searched,
!     and read the diagonal and off-diagonal elements of the matrix
!     A from data file

      Read (nin,*) vl, vu
      Read (nin,*) d(1:n)
```

```
      Read (nin,*) e(1:n-1)

!     Set the absolute error tolerance for eigenvalues.  With ABSTOL
!     set to zero, the default value would be used instead

      abstol = 2.0E0_nag_wp*x02amf()

!     Solve the symmetric eigenvalue problem
!     The NAG name equivalent of dstevx is f08jbf
      Call dstevx('Vectors','Values in range',n,d,e,vl,vu,il,iu,abstol,m,w,z, &
        ldz,work,iwork,jfail,info)

      If (info>=0) Then

!       Print solution

        Write (nout,99999) 'Number of eigenvalues found =', m
        Write (nout,*)
        Write (nout,*) 'Eigenvalues'
        Write (nout,99998) w(1:m)
        Flush (nout)

!       ifail: behaviour on error exit
!              =0 for hard exit, =1 for quiet-soft, =-1 for noisy-soft
        ifail = 0
        Call x04caf('General',' ',n,m,z,ldz,'Selected eigenvectors',ifail)

        If (info>0) Then
          Write (nout,99999) 'INFO eigenvectors failed to converge, INFO =', &
            info
          Write (nout,*) 'Indices of eigenvectors that did not converge'
          Write (nout,99997) jfail(1:m)
        End If
      Else
        Write (nout,99999) 'Failure in DSTEVX. INFO =', info
      End If

99999 Format (1X,A,I5)
99998 Format (3X,(8F8.4))
99997 Format (3X,(8I8))
    End Program f08jbfe
```

## 9.2 Program Data

```
F08JBF Example Program Data

  4                    :Value of N

  0.0  5.0             :Values of VL and VU

  1.0  4.0  9.0  16.0  :End of diagonal elements
  1.0  2.0  3.0        :End of off-diagonal elements
```

## 9.3 Program Results

```
 F08JBF Example Program Results

 Number of eigenvalues found =    2

 Eigenvalues
    0.6476  3.5470
 Selected eigenvectors
          1       2
 1   0.9396  0.3388
 2  -0.3311  0.8628
 3   0.0853 -0.3648
 4  -0.0167  0.0879
```